



**WEB SÉMANTIQUE ET ONTOLOGIES**

**WEB DES DONNÉES**

**DONNÉES LIÉES (LINKED DATA)**

## **4 – INTERROGER LES DONNÉES AVEC SPARQL**

Philippe GENOUD – Danielle ZIEBELIN – Mohamed Wadhah Mabrouk

LIG


[Prenom.Nom@imag.fr](mailto:Prenom.Nom@imag.fr)



# Querying Linked Data with SPARQL

## Linked Data: 3<sup>rd</sup> Principle

*When someone looks up a URI, provide useful information, using the standards (RDF, SPARQL).*



Most apps use only a subset of the stack


Querying allows fine-grained data access

Standardized information exchange is key

Formats are necessary, but not too important

The Semantic Web is based on the Web

Linked Data uses a small selection of technologies



SPARQL Protocol And RDF Query Language

## Exercice SPARQL

- "DBpedia est un projet universitaire et communautaire d'exploration et extraction automatiques de données dérivées de Wikipédia. Son principe est de proposer une version structurée et sous forme de données normalisées au format du web sémantique des contenus encyclopédiques de chaque fiche encyclopédique. DBpedia vise aussi à relier à Wikipédia (et inversement) des ensembles d'autres données ouvertes provenant du Web des données." (<http://fr.wikipedia.org/wiki/DBpedia>).
- L'url `http://dbpedia.org/sparql` correspond à un point d'accès SPARQL (SPARQL endpoint) aux données de DBPedia. Vous allez utiliser celui-ci pour explorer les données RDF de DBpedia.

# Dbpedia sparql endpoint

Interface web du point d'accès SPARQL de DBPedia : <https://dbpedia.org/sparql>

SPARQL Query Editor

About

Tables

Conductor

Facet Browser

Permalink

Extensions: cxml save to dav sponge User: SPARQL

Default Data Set Name (Graph IRI)

<http://dbpedia.org>

Query Text

```
select distinct ?Concept where {[] a ?Concept} LIMIT 100
```

Results Format

HTML

Execute Query

Reset

Execution timeout

30000

milliseconds

Options

☒ Strict checking of void variables

☒ Strict checking of variable names used in multiple clauses but not logically connected to each other

☐ Suppress errors on wrong geometries and errors on geometrical operators (failed operations will return NULL)

☐ Log debug info at the end of output (has no effect on some queries and output formats)

☐ Generate SPARQL compilation report (instead of executing the query)

Copyright © 2023 OpenLink Software

Virtuoso version 08.03.3329 (42780c0b7c) on Linux (x86\_64-generic-linux-glibc25) Single Server Edition (61 GB total memory, 43 GB memory in use)

# Question 1

- Question 1: Dans la zone texte prévue à cet effet tapez la requête SPARQL suivante :

```
SELECT ?m ?a WHERE {  
  ?m a dbo:Mountain;  
    dbo:locatedInArea dbr:Nepal;  
    dbo:elevation ?a.  
}
```

- a) D'après vous que fait cette requête?
- b) à quoi correspond le a dans le pattern ?m a dbo:Mountain; ? A quelle URI se substitue-t-il ?
- c) A quelles URIs correspondent les préfixes dbo:, dbr: ?
- d) Exécutez la requête et observez le résultat obtenu. Correspond-t-il à ce que vous aviez prévu ?

## Question 1: solution 1)

- Question 1: Dans la zone texte prévue à cet effet tapez la requête SPARQL suivante :

```
SELECT ?m ?a WHERE {  
  ?m a dbo:Mountain;  
    dbo:locatedInArea dbr:Nepal;  
    dbo:elevation ?a.  
}
```

- D'après vous que fait cette requête?

***Cette requête SPARQL recherche dans le graphe DBpedia toutes les ressources de type Montagne situées au Népal qui ont une altitude.***

***Le resultat produit est l'ensemble des bindings qui répondent à cette question, où ?m est l'URI des ressources trouvées et ?a est l'altitude de celles-ci.***

## Question 1: solution 2)

- à quoi correspond le **a** dans le pattern ?m **a** dbo:Mountain; ? A quelle URI se substitue-t-il ?

***a est un raccourci que proposent Turtle et SPARQL en lieu et place de la propriété type définie dans le vocabulaire RDF.***

***l'URI de celle-ci est <http://www.w3.org/1999/02/22-rdf-syntax-ns#type> que l'on abrège souvent en `rdf:type` ou `rdf:` est le préfixe correspondant à <http://www.w3.org/1999/02/22-rdf-syntax-ns#>.***

## Question 1: solution 4)

- A quelles URIs correspondent les préfixes dbo:, dbr: ?

Les préfixes dbo: et dbr: sont implicites dans l'interface web du SPARQL-endpoint de DBpedia. La liste ci-dessous présente à quelles URIs il correspondent

préfixe URI dbo: c'est le préfixe de l'ontologie (le vocabulaire : types et propriétés) de dbpedia  
<http://dbpedia.org/ontology/>


préfixe URI dbr: c'est le préfixe pour les ressources de dbpedia <http://dbpedia.org/resource/>

On peut trouver la liste des préfixes utilisés implicitement par le SPARQL endpoint de DBpedia en cliquant sur le lien de navigation correspondant dans l'interface web.



## Question 1: solution 3)

On peut trouver la liste des préfixes utilisés implicitement par le SPARQL endpoint de DBpedia en cliquant sur le lien de navigation correspondant dans l'interface web.



Et en cliquant sur Namespace

# Question 1: solution 3)

Extrait des namespaces

SPARQL Query Editor

## Namespace Prefixes

Prefix	URI
a	<a href="http://www.w3.org/2005/Atom">http://www.w3.org/2005/Atom</a>
address	<a href="http://schemas.talis.com/2005/address/schema#">http://schemas.talis.com/2005/address/schema#</a>
admin	<a href="http://webns.net/mvcb/">http://webns.net/mvcb/</a>
as	<a href="https://www.w3.org/ns/activitystreams#">https://www.w3.org/ns/activitystreams#</a>
atom	<a href="http://atomowl.org/ontologies/atomrdf#">http://atomowl.org/ontologies/atomrdf#</a>
aws	<a href="http://soap.amazon.com/">http://soap.amazon.com/</a>
b3s	<a href="http://b3s.openlinksw.com/">http://b3s.openlinksw.com/</a>
batch	<a href="http://schemas.google.com/gdata/batch">http://schemas.google.com/gdata/batch</a>
bibo	<a href="http://purl.org/ontology/bibo/">http://purl.org/ontology/bibo/</a>
bif	<a href="http://www.openlinksw.com/schemas/bif#">http://www.openlinksw.com/schemas/bif#</a>
bugzilla	<a href="http://www.openlinksw.com/schemas/bugzilla#">http://www.openlinksw.com/schemas/bugzilla#</a>
c	<a href="http://www.w3.org/2002/12/cal/icaltzd#">http://www.w3.org/2002/12/cal/icaltzd#</a>
cb	<a href="http://www.crunchbase.com/">http://www.crunchbase.com/</a>
cc	<a href="http://web.resource.org/cc/">http://web.resource.org/cc/</a>
content	<a href="http://purl.org/rss/1.0/modules/content/">http://purl.org/rss/1.0/modules/content/</a>
cv	<a href="http://purl.org/captsolo/resume-rdf/0.2/cv#">http://purl.org/captsolo/resume-rdf/0.2/cv#</a>
cvbase	<a href="http://purl.org/captsolo/resume-rdf/0.2/base#">http://purl.org/captsolo/resume-rdf/0.2/base#</a>

## Question 1: solution 3)

Extrait des namespaces

Préfixe	Namespace	Description
dbo:	<a href="http://dbpedia.org/ontology/">http://dbpedia.org/ontology/</a>	Ontologie DBpedia (concepts, propriétés)
dbr:	<a href="http://dbpedia.org/resource/">http://dbpedia.org/resource/</a>	Ressources concrètes (instances Wikipedia)
dbp:	<a href="http://dbpedia.org/property/">http://dbpedia.org/property/</a>	Propriétés extraites directement de Wikipedia infoboxes
yago:	<a href="http://dbpedia.org/class/yago/">http://dbpedia.org/class/yago/</a>	Classes issues du projet YAGO
foaf:	<a href="http://xmlns.com/foaf/0.1/">http://xmlns.com/foaf/0.1/</a>	Données sur les personnes
rdfs:	<a href="http://www.w3.org/2000/01/rdf-schema#">http://www.w3.org/2000/01/rdf-schema#</a>	Schéma RDF standard
rdf:	<a href="http://www.w3.org/1999/02/22-rdf-syntax-ns#">http://www.w3.org/1999/02/22-rdf-syntax-ns#</a>	Syntaxe RDF
owl:	<a href="http://www.w3.org/2002/07/owl#">http://www.w3.org/2002/07/owl#</a>	Ontology Web Language
xsd:	<a href="http://www.w3.org/2001/XMLSchema#">http://www.w3.org/2001/XMLSchema#</a>	Types de données

## Question 1: solution 3)

On peut aussi les trouver à l'aide du service web  
prefix.cc ( <https://prefix.cc/>)


La requête SPARQL complète incluant les prefixes  
serait :

```
PREFIX dbo <http://dbpedia.org/ontology/>
PREFIX dbr: <http://dbpedia.org/resource/>
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-
syntax-ns#>
SELECT ?m ?a WHERE {
    ?m rdf:type dbo:Mountain;
        dbo:locatedInArea dbr:Nepal;
        dbo:elevation ?a.
}
```

## Question 1: solution 4)

- Exécutez la requête et observez le résultat obtenu. Correspond-t-il à ce que vous aviez prévu ?

```
SELECT ?m ?a WHERE {  
  ?m a dbo:Mountain;  
  dbo:locatedInArea dbr:Nepal;  
  dbo:elevation ?a.  
}
```



m	a
<a href="http://dbpedia.org/resource/Pumori">http://dbpedia.org/resource/Pumori</a>	7161.0
<a href="http://dbpedia.org/resource/Purbung">http://dbpedia.org/resource/Purbung</a>	6465.0
<a href="http://dbpedia.org/resource/Putha_Hiunchuli">http://dbpedia.org/resource/Putha_Hiunchuli</a>	7246.0
<a href="http://dbpedia.org/resource/Sandakphu">http://dbpedia.org/resource/Sandakphu</a>	3636.0
<a href="http://dbpedia.org/resource/Ombigaichan">http://dbpedia.org/resource/Ombigaichan</a>	6340.0
<a href="http://dbpedia.org/resource/Baruntse">http://dbpedia.org/resource/Baruntse</a>	7162.0
<a href="http://dbpedia.org/resource/Hiunchuli">http://dbpedia.org/resource/Hiunchuli</a>	6441.0
<a href="http://dbpedia.org/resource/Paungda_Danda">http://dbpedia.org/resource/Paungda_Danda</a>	4755.0
<a href="http://dbpedia.org/resource/Saipal">http://dbpedia.org/resource/Saipal</a>	7031.0

## Question 2 et 3:

- 1) A quoi correspond l'url <http://dbpedia.org/resource/Saipal> qui apparaît dans la colonne m du tableau résultat ?
- 2) Cliquez sur ce lien. A quoi correspond le résultat obtenu ?
- 3) Le lien était bien <https://dbpedia.org/resource/Saipal> (vous pouvez le vérifier en tapant directement cette url dans la barre d'adresse de votre navigateur), pourtant l'url de la page affichée est <https://dbpedia.org/page/Saipal>.
  - Que s'est-il passé ? Pouvez-vous expliquer ce comportement de votre navigateur ?
  - Que faudrait-il modifier dans la requête HTTP pour obtenir en RDF les données de DBpedia concernant Saipal ?
  - Récupérez ces données au format Turtle. Expliquez comment vous avez procédé.
- Donnez une requête SPARQL qui permet d'obtenir les URIs et l'altitude des dix plus hauts sommets du Népal.

## Question 2: solution 1)

- a) A quoi correspond l'url `http://dbpedia.org/resource/Saipal` qui apparait dans la colonne m du tableau résultat ?

A chaque page Wikipedia dont l'URL de la forme <https://en.wikipedia.org/wiki/XXXX> correspond une ressource DBpedia dont l'URI est `http://dbpedia.org/resource/XXXX` où XXXX désigne le sujet de la page. A la page consacrée à Saipal (`https://en.wikipedia.org/wiki/Saipal`) correspond la ressource `http://dbpedia.org/resource/Saipal`.

## Question 2: solution 2)

- 2) Cliquez sur ce lien. A quoi correspond le résultat obtenu ?

La page affichée dans le navigateur représente sous une forme lisible par un humain (texte HTML), l'ensemble des triplets dont la ressource <http://dbpedia.org/resource/Saipal> est le sujet.

Lorsque l'objet du triplet est une ressource DBpedia, le lien hypertexte permet de naviguer vers la description de celle-ci, en d'autres termes de parcourir le graphe RDF.



## Question 2: solution 2)



Browse using ▾ Formats ▾

Faceted Browser Sparql Endpoint

### About: [Saipal](#)

An Entity of Type: [lieu](#), from Named Graph: <http://dbpedia.org>, within Data Space: [dbpedia.org](#)

Saipal is a mountain in the Himalayas of north-west Bajhang district in Nepal. Api, Nampa and Saipal are a trio of high mountains located in northwestern Nepal. Together they form a small range of sharp, icy peaks, rising from a long, steep, snowy crest.

Property	Value
<a href="#">dbo:abstract</a>	<ul style="list-style-type: none"><li>Saipal is a mountain in the Himalayas of north-west Bajhang district in Nepal. Api, Nampa and Saipal are a trio of high mountains located in northwestern Nepal. Together they form a small range of sharp, icy peaks, rising from a long, steep, snowy crest. (en)</li></ul>
<a href="#">dbo:elevation</a>	<ul style="list-style-type: none"><li>7031.000000 (xsd:double)</li></ul>
<a href="#">dbo:firstAscentYear</a>	<ul style="list-style-type: none"><li>1960-01-01 (xsd:gYear)</li></ul>
<a href="#">dbo:locatedInArea</a>	<ul style="list-style-type: none"><li><a href="#">dbr:Sudurpashchim_Province</a></li><li><a href="#">dbr:Nepal</a></li></ul>
<a href="#">dbo:prominence</a>	<ul style="list-style-type: none"><li>1824.000000 (xsd:double)</li></ul>
<a href="#">dbo:wikiPageID</a>	<ul style="list-style-type: none"><li>12850731 (xsd:integer)</li></ul>
<a href="#">dbo:wikiPageLength</a>	<ul style="list-style-type: none"><li>1476 (xsd:nonNegativeInteger)</li></ul>

## Question 2: solution 3)

- 3) Le lien était bien `https://dbpedia.org/resource/Saipal` (vous pouvez le vérifier en tapant directement cette url dans la barre d'adresse de votre navigateur), pourtant l'url de la page affichée est `https://dbpedia.org/page/Saipal`.
  - Que s'est-il passé ? Pouvez-vous expliquer ce comportement de votre navigateur ?

**L'URL affiché dans la barre de navigation est :**

**`http://dbpedia.org/page/Saipal`**

Dans l'URI de la ressource, `resource` a été remplacé par `page`. Elle diffère de celle de la ressource car il y a eu négociation de contenu entre le navigateur et le serveur web de dbpedia. Le navigateur a envoyé un en-tête HTTP `Accept-Content : text/html`. Le serveur a répondu en faisant une redirection vers l'URL <http://dbpedia.org/page/Saipal> pour obtenir une page HTML décrivant la ressource.

## Question 2: solution 3)

- Que faudrait-il modifier dans la requête HTTP pour obtenir en RDF les données de DBpedia concernant Saipal ?

**Pour récupérer une description en Turtle il suffit de faire une requête HTTP de type GET avec l'url de la ressource**  
([http://dbpedia.org/resource/Grenoble\\_Alpes\\_University](http://dbpedia.org/resource/Grenoble_Alpes_University)) mais en précisant dans l'en tête Accept le type MIME text/turtle.

- Récupérez ces données au format Turtle. Expliquez comment vous avez procédé.

**Pour récupérer la réponse en Turtle, plusieurs moyens sont possibles :**

**Utiliser l'onglet Network des outils de développement du navigateur et renvoyer la requête en modifiant l'en-tête (header) Accept**

**Utiliser un outil ligne de commande wget ou curl**

## Question 3: solution

- Donnez une requête SPARQL qui permet d'obtenir les URIs et l'altitude des dix plus hauts sommets du Népal.

**EN SPARQL les graph patterns d'une requête SELECT génèrent une collection non ordonnée de solutions (solution sequence). Cette séquence peut être modifiée en utilisant les modifieurs de séquence définis dans le langage SPARQL. Pour savoir quels modifieurs sont disponibles et comment les utiliser, reportez vous à la documentation de référence de SPARQL.**

**Pour ordonner les résultats par ordre décroissant on utilise le modifieur ORDER BY et le modifieur LIMIT pour restreindre à 10 le nombre de solutions.**

```
select ?m ?a where {  
    ?m rdf:type dbo:Mountain;  
    dbo:locatedInArea dbr:Nepal;  
    dbo:elevation ?a.  
} ORDER by DESC(?a) LIMIT 10
```

## Question 3: solution

Vorkbench Sign in · GitLab MIAI Grenoble Alpes -... VocBench dblp: computer scienc... Intranet de l'Université... >> ☐ Autres marque-

< SPARQL | HTML5 table

m	a
<a href="http://dbpedia.org/resource/Kangchenjunga">http://dbpedia.org/resource/Kangchenjunga</a>	8586.0
<a href="http://dbpedia.org/resource/Lhotse">http://dbpedia.org/resource/Lhotse</a>	8516.0
<a href="http://dbpedia.org/resource/Makalu">http://dbpedia.org/resource/Makalu</a>	8481.0
<a href="http://dbpedia.org/resource/Lhotse_Middle">http://dbpedia.org/resource/Lhotse_Middle</a>	8410.0
<a href="http://dbpedia.org/resource/Lhotse_Shar">http://dbpedia.org/resource/Lhotse_Shar</a>	8383.0
<a href="http://dbpedia.org/resource/Cho_Oyu">http://dbpedia.org/resource/Cho_Oyu</a>	8188.0
<a href="http://dbpedia.org/resource/Dhaulagiri">http://dbpedia.org/resource/Dhaulagiri</a>	8167.0
<a href="http://dbpedia.org/resource/Manaslu">http://dbpedia.org/resource/Manaslu</a>	8163.0
<a href="http://dbpedia.org/resource/Annapurna">http://dbpedia.org/resource/Annapurna</a>	8091.0
<a href="http://dbpedia.org/resource/Gyachung_Kang">http://dbpedia.org/resource/Gyachung_Kang</a>	7952.0

## Question 4

- Complétez la requête précédente pour obtenir en plus de l'altitude, le nom de chaque sommet défini par la valeur de la propriété `rdfs:label`.
- Qu'observez-vous ? Combien de sommets différents obtenez-vous ? Pourquoi ?
- Modifiez la requête pour faire en sorte de n'obtenir que les 10 plus hauts sommets du Népal avec uniquement leur nom en anglais.

## Question 4 : solution

- Complétez la requête précédente pour obtenir en plus de l'altitude, le nom de chaque sommet défini par la valeur de la propriété `rdfs:label`.

Pour obtenir les labels des sommets, il faut dans la requête précédente rejouter le pattern `?m rdfs:label ?l`. Ce qui donne la requête ci-dessous :

```
select ?m ?a ?l where {  
    ?m rdf:type dbo:Mountain;  
    dbo:locatedInArea dbr:Nepal;  
    dbo:elevation ?a;  
    rdfs:label ?l.  
} ORDER by DESC(?a) LIMIT 10
```

## Question 4 : solution

SPARQL | HTML5 table

m	a	l
<a href="http://dbpedia.org/resource/Kangchenjunga">http://dbpedia.org/resource/Kangchenjunga</a>	8586.0	"Kangchenjunga"@en
<a href="http://dbpedia.org/resource/Kangchenjunga">http://dbpedia.org/resource/Kangchenjunga</a>	8586.0	"كانغشينجونا"@ar
<a href="http://dbpedia.org/resource/Kangchenjunga">http://dbpedia.org/resource/Kangchenjunga</a>	8586.0	"Kanchenjunga"@ca
<a href="http://dbpedia.org/resource/Kangchenjunga">http://dbpedia.org/resource/Kangchenjunga</a>	8586.0	"Kančendženga"@cs
<a href="http://dbpedia.org/resource/Kangchenjunga">http://dbpedia.org/resource/Kangchenjunga</a>	8586.0	"Kangchendzōnga"@de
<a href="http://dbpedia.org/resource/Kangchenjunga">http://dbpedia.org/resource/Kangchenjunga</a>	8586.0	"Κανγκτσενγχιούνγκα"@el
<a href="http://dbpedia.org/resource/Kangchenjunga">http://dbpedia.org/resource/Kangchenjunga</a>	8586.0	"Kangchenjunga"@eo
<a href="http://dbpedia.org/resource/Kangchenjunga">http://dbpedia.org/resource/Kangchenjunga</a>	8586.0	"Kanchenjunga"@eu
<a href="http://dbpedia.org/resource/Kangchenjunga">http://dbpedia.org/resource/Kangchenjunga</a>	8586.0	"Kanchenjunga"@es
<a href="http://dbpedia.org/resource/Kangchenjunga">http://dbpedia.org/resource/Kangchenjunga</a>	8586.0	"Kangchenjunga"@fr



## Question 4 : solution

- Qu'observez-vous ? Combien de sommets différents obtenez-vous ? Pourquoi ?

Cependant on n'obtient qu'un sommet: le Kangchenjunga

En effet pour chaque sommet plusieurs labels sont disponibles dans des langues différentes, un même sommet apparaît donc dans autant de binding qu'il a de labels différents.

## Question 4 : solution

- Modifiez la requête pour faire en sorte de n'obtenir que les 10 plus hauts sommets du Népal avec uniquement leur nom en anglais.

Un filtre est une clause que l'on peut insérer dans une requête SPARQL pour, comme son nom l'indique, filtrer les résultats. Dans un filtre une expression booléenne est définie et seuls les résultats pour lesquels l'expression est vraie sont conservés. Hint : fonctions filtres sur les chaînes.

```
select ?m ?a ?l where {  
    ?m rdf:type dbo:Mountain;  
    dbo:locatedInArea dbr:Nepal;  
    dbo:elevation ?a;  
    rdfs:label ?l.  
    FILTER langMatches( lang(?l), "EN" )  
} ORDER by DESC(?a) LIMIT 10
```

## Question 4 : solution

SPARQL | HTML5 table

m	a	l
<a href="http://dbpedia.org/resource/Kangchenjunga">http://dbpedia.org/resource/Kangchenjunga</a>	8586.0	"Kangchenjunga"@en
<a href="http://dbpedia.org/resource/Lhotse">http://dbpedia.org/resource/Lhotse</a>	8516.0	"Lhotse"@en
<a href="http://dbpedia.org/resource/Makalu">http://dbpedia.org/resource/Makalu</a>	8481.0	"Makalu"@en
<a href="http://dbpedia.org/resource/Lhotse_Middle">http://dbpedia.org/resource/Lhotse_Middle</a>	8410.0	"Lhotse Middle"@en
<a href="http://dbpedia.org/resource/Lhotse_Shar">http://dbpedia.org/resource/Lhotse_Shar</a>	8383.0	"Lhotse Shar"@en
<a href="http://dbpedia.org/resource/Cho_Oyu">http://dbpedia.org/resource/Cho_Oyu</a>	8188.0	"Cho Oyu"@en
<a href="http://dbpedia.org/resource/Dhaulagiri">http://dbpedia.org/resource/Dhaulagiri</a>	8167.0	"Dhaulagiri"@en
<a href="http://dbpedia.org/resource/Manaslu">http://dbpedia.org/resource/Manaslu</a>	8163.0	"Manaslu"@en
<a href="http://dbpedia.org/resource/Annapurna">http://dbpedia.org/resource/Annapurna</a>	8091.0	"Annapurna"@en
<a href="http://dbpedia.org/resource/Gyachung_Kang">http://dbpedia.org/resource/Gyachung_Kang</a>	7952.0	"Gyachung Kang"@en

## Question 5 :

- La requête précédente donne-t-elle tous les sommets du Népal de plus de 8000m ?  
  
a) Modifiez la requête de façon à obtenir dans l'ordre du plus haut au plus bas, l'altitude et le nom anglais de tous les sommets népalais de plus de 8000m .  
  
b) Combien de sommets trouvez vous ? Ecrivez un requête qui vous donne directement ce résultat (le nombre de sommets népalais de plus de 8000 m).

## Question 5 : solution 1)

- La requête précédente donne-t-elle tous les sommets du Népal de plus de 8000m ?  
1) Modifiez la requête de façon à obtenir dans l'ordre du plus haut au plus bas, l'altitude et le nom anglais de tous les sommets népalais de plus de 8000m .

**Pour obtenir tous les sommets de plus de 8000 (qui sont neuf), il faut dans la requête précédente rejouter un filtre sur l'altitude et supprimer le modifieur limitant aux 10 premiers résultats. Ce qui donne la requête ci-dessous :**

```
select ?m ?a ?l where {  
  ?m rdf:type dbo:Mountain;  
    dbo:locatedInArea dbr:Nepal;  
    dbo:elevation ?a;  
    rdfs:label ?l.  
  FILTER langMatches( lang(?l), "EN" )  
  FILTER (?a >= 8000)  
} ORDER by DESC(?a)
```

## Question 5 : solution 1)

SPARQL | HTML5 table

m	a	l
<a href="http://dbpedia.org/resource/Kangchenjunga">http://dbpedia.org/resource/Kangchenjunga</a>	8586.0	"Kangchenjunga"@en
<a href="http://dbpedia.org/resource/Lhotse">http://dbpedia.org/resource/Lhotse</a>	8516.0	"Lhotse"@en
<a href="http://dbpedia.org/resource/Makalu">http://dbpedia.org/resource/Makalu</a>	8481.0	"Makalu"@en
<a href="http://dbpedia.org/resource/Lhotse_Middle">http://dbpedia.org/resource/Lhotse_Middle</a>	8410.0	"Lhotse Middle"@en
<a href="http://dbpedia.org/resource/Lhotse_Shar">http://dbpedia.org/resource/Lhotse_Shar</a>	8383.0	"Lhotse Shar"@en
<a href="http://dbpedia.org/resource/Cho_Oyu">http://dbpedia.org/resource/Cho_Oyu</a>	8188.0	"Cho Oyu"@en
<a href="http://dbpedia.org/resource/Dhaulagiri">http://dbpedia.org/resource/Dhaulagiri</a>	8167.0	"Dhaulagiri"@en
<a href="http://dbpedia.org/resource/Manaslu">http://dbpedia.org/resource/Manaslu</a>	8163.0	"Manaslu"@en
<a href="http://dbpedia.org/resource/Annapurna">http://dbpedia.org/resource/Annapurna</a>	8091.0	"Annapurna"@en


## Question 5 : solution 2)

b) Combien de sommets trouvez vous ? Ecrivez un requête qui vous donne directement ce résultat (le nombre de sommets népalais de plus de 8000 m).

**hint: utilisez la fonction d'aggrégation COUNT.**

**Pour obtenir le nombre de sommets de plus de 8000 (qui sont neuf), il faut utiliser le filtre COUNT de la manière suivante :**

```
select (COUNT(?m) AS ?nb8000) where {  
  ?m rdf:type dbo:Mountain;  
  dbo:locatedInArea dbr:Nepal;  
  dbo:elevation ?a.  
  FILTER (?a >= 8000)  
}
```



SPARQL | HTML5 table

nb8000
9

## Question 6

- Si tous les sommets népalais de plus de 8000 m ont un nom en anglais, DBpedia ne contient le nom en russe que de certains d'entre eux.
- Pouvez-vous dire combien et lesquels ? Indiquez la ou les requêtes qui vous a permis de répondre à ces questions.
- Ecrivez une requête qui pour ces sommets donne simultanément leur URI, leur altitude, leur nom en anglais et leur nom en russe et les trie selon leur altitude (l'ordre décroissant).
- Inversement donnez une requête qui donne la liste des sommets de plus de 8000 qui n'ont pas de nom (label) en russe
- Donnez une requête qui donne la liste des tous les sommets de plus de 8000 avec leur altitude, leur nom en anglais et **si il existe** leur nom en russe.




## Question 6 : solution

- Si tous les sommets népalais de plus de 8000m ont un nom en anglais, DBpedia ne contient le nom en russe que de certains d'entre eux.
- Pouvez-vous dire combien et lesquels ? Indiquez la ou les requêtes qui vous a permis de répondre à ces questions.

**Pour obtenir le nombre de sommets de plus de 8000 du Népal dont DBpedia connaît le nom en russe, on peut utiliser la requête suivante :**

```
SELECT (COUNT(?m) AS ?nb8000) WHERE {  
  ?m rdf:type dbo:Mountain;  
    dbo:locatedInArea dbr:Nepal;  
    dbo:elevation ?a;  
    rdfs:label ?l  
  FILTER langMatches( lang(?l), "RU" )  
  FILTER (?a >= 8000)  
}
```



SPARQL | HTML5 table

nb8000
5

## Question 6 : solution

- Pouvez-vous dire lesquels ?

Pour connaitre quels sont ces sommets on pourrait utiliser la requête suivante :

```
SELECT ?m ?a ?l WHERE {  
  ?m rdf:type dbo:Mountain;  
    dbo:locatedInArea dbr:Nepal;  
    dbo:elevation ?a;  
    rdfs:label ?l  
  FILTER langMatches( lang(?l), "RU" )  
  FILTER (?a >= 8000)  
}
```

## Question 6 : solution

SPARQL | HTML5 table

m	a	l
<a href="http://dbpedia.org/resource/Lhotse">http://dbpedia.org/resource/Lhotse</a>	8516.0	"Лхоцзе"@ru
<a href="http://dbpedia.org/resource/Makalu">http://dbpedia.org/resource/Makalu</a>	8481.0	"Макалу"@ru
<a href="http://dbpedia.org/resource/Manaslu">http://dbpedia.org/resource/Manaslu</a>	8163.0	"Манаслу"@ru
<a href="http://dbpedia.org/resource/Cho_Oyu">http://dbpedia.org/resource/Cho_Oyu</a>	8188.0	"Чо-Ойю"@ru
<a href="http://dbpedia.org/resource/Kangchenjunga">http://dbpedia.org/resource/Kangchenjunga</a>	8586.0	"Канченджанга"@ru

## Question 6 : solution

- Ecrivez une requête qui pour ces sommets donne simultanément leur URI, leur altitude, leur nom en anglais et leur nom en russe et les trie selon leur altitude (l'ordre décroissant).

Pour obtenir les noms anglais et russe des sommets de plus de 8000 du Népal dont DBpedia connaît le nom en russe, on peut utiliser la requête suivante :

```
SELECT ?m ?a ?l1 ?l2 WHERE {  
    ?m rdf:type dbo:Mountain;  
        dbo:locatedInArea dbr:Nepal;  
        dbo:elevation ?a;  
        rdfs:label ?l1, ?l2 .  
    FILTER (langMatches( lang(?l1), "EN" ) &&  
langMatches( lang(?l2), "RU" ))  
    FILTER (?a >= 8000)  
} ORDER BY DESC(?a)
```

## Question 6 : solution

SPARQL | HTML5 table

m	a	l1	l2
<a href="http://dbpedia.org/resource/Kangchenjunga">http://dbpedia.org/resource/Kangchenjunga</a>	8586.0	"Kangchenjunga"@en	"Канченджанга"@ru
<a href="http://dbpedia.org/resource/Lhotse">http://dbpedia.org/resource/Lhotse</a>	8516.0	"Lhotse"@en	"Лхоцзе"@ru
<a href="http://dbpedia.org/resource/Makalu">http://dbpedia.org/resource/Makalu</a>	8481.0	"Makalu"@en	"Макалу"@ru
<a href="http://dbpedia.org/resource/Cho_Oyu">http://dbpedia.org/resource/Cho_Oyu</a>	8188.0	"Cho Oyu"@en	"Чо-Ойю"@ru
<a href="http://dbpedia.org/resource/Manaslu">http://dbpedia.org/resource/Manaslu</a>	8163.0	"Manaslu"@en	"Манаслу"@ru

## Question 6 : solution

- Inversement donnez une requête qui donne la liste des sommets de plus de 8000 qui n'ont pas de nom (label) en russe

Utilisez la négation et la possibilité d'effectuer un filtrage en testant l'absence d'un pattern. Negation dans SPARQL.

Pour obtenir les noms anglais des sommets de plus de 8000 du Népal pour lesquels DBpedia n'a pas de nom en russe, on peut utiliser la requête suivante :

```
SELECT ?m ?a ?l1 WHERE {  
    ?m rdf:type dbo:Mountain;  
    dbo:locatedInArea dbr:Nepal;  
    dbo:elevation ?a;  
    rdfs:label ?l1 .  
    FILTER (langMatches( lang(?l1), "EN" ))  
    FILTER (NOT EXISTS { ?m rdfs:label ?l2. FILTER  
(langMatches( lang(?l2), "RU")) })  
    FILTER (?a >= 8000)  
} ORDER BY DESC(?a)
```

## Question 6 : solution

SPARQL | HTML5 table

m	a	l1
<a href="http://dbpedia.org/resource/Lhotse_Middle">http://dbpedia.org/resource/Lhotse_Middle</a>	8410.0	"Lhotse Middle"@en
<a href="http://dbpedia.org/resource/Lhotse_Shar">http://dbpedia.org/resource/Lhotse_Shar</a>	8383.0	"Lhotse Shar"@en
<a href="http://dbpedia.org/resource/Dhaulagiri">http://dbpedia.org/resource/Dhaulagiri</a>	8167.0	"Dhaulagiri"@en
<a href="http://dbpedia.org/resource/Annapurna">http://dbpedia.org/resource/Annapurna</a>	8091.0	"Annapurna"@en

## Question 6 : solution

- Donnez une requête qui donne la liste des tous les sommets de plus de 8000 avec leur altitude, leur nom en anglais et **si il existe** leur nom en russe.

**Pour obtenir les noms anglais et éventuellement russes des sommets de plus de 8000 du Népal on peut utiliser la requête suivante :**

```
SELECT ?m ?a ?l1 ?l2 WHERE {  
    ?m rdf:type dbo:Mountain;  
        dbo:locatedInArea dbr:Nepal;  
        dbo:elevation ?a;  
        rdfs:label ?l1 .  
    FILTER (?a >= 8000)  
    FILTER (langMatches( lang(?l1), "EN" ))  
    OPTIONAL {  
        ?m rdfs:label ?l2.  
        FILTER (langMatches( lang(?l2), "RU" ))  
    }  
} ORDER BY DESC(?a)
```



## Question 6 : solution

SPARQL   HTML5 table			
m	a	l1	l2
<a href="http://dbpedia.org/resource/Kangchenjunga">http://dbpedia.org/resource/Kangchenjunga</a>	8586.0	"Kangchenjunga"@en	"Канченджанга"@ru
<a href="http://dbpedia.org/resource/Lhotse">http://dbpedia.org/resource/Lhotse</a>	8516.0	"Lhotse"@en	"Лхоцзе"@ru
<a href="http://dbpedia.org/resource/Makalu">http://dbpedia.org/resource/Makalu</a>	8481.0	"Makalu"@en	"Макалу"@ru
<a href="http://dbpedia.org/resource/Lhotse_Middle">http://dbpedia.org/resource/Lhotse_Middle</a>	8410.0	"Lhotse Middle"@en	
<a href="http://dbpedia.org/resource/Lhotse_Shar">http://dbpedia.org/resource/Lhotse_Shar</a>	8383.0	"Lhotse Shar"@en	
<a href="http://dbpedia.org/resource/Cho_Oyu">http://dbpedia.org/resource/Cho_Oyu</a>	8188.0	"Cho Oyu"@en	"Чо-Ойю"@ru
<a href="http://dbpedia.org/resource/Dhaulagiri">http://dbpedia.org/resource/Dhaulagiri</a>	8167.0	"Dhaulagiri"@en	
<a href="http://dbpedia.org/resource/Manaslu">http://dbpedia.org/resource/Manaslu</a>	8163.0	"Manaslu"@en	"Манаслу"@ru
<a href="http://dbpedia.org/resource/Annapurna">http://dbpedia.org/resource/Annapurna</a>	8091.0	"Annapurna"@en	

## Question 7

- Donnez une requête qui vous permet de trouver les sommets dont la première ascension a été effectuée par Lionel Terray, l'année de cette première ascension et si DBpedia dispose des données le nom du ou des alpinistes avec qui Lionel Terray a accompli cette première.

Vous devriez obtenir un résultat ressemblant à celui obtenu dans l'image ci-dessous

SPARQL | HTML5 table

m	nom	altitude	pnom	année
<a href="http://dbpedia.org/resource/Makalu">http://dbpedia.org/resource/Makalu</a>	"Makalu"@en	8481.0	"Jean Couzy"@en	"1955"^^<http://www.w3.org/2001/XMLSchema#gYear>
<a href="http://dbpedia.org/resource/Chomo_Lonzo">http://dbpedia.org/resource/Chomo_Lonzo</a>	"Chomo Lonzo"@en	7804.0	"Jean Couzy"@en	"1954"^^<http://www.w3.org/2001/XMLSchema#gYear>
<a href="http://dbpedia.org/resource/Kumbhakarna_Mountain">http://dbpedia.org/resource/Kumbhakarna_Mountain</a>	"Kumbhakarna Mountain"@en	7710.0		"0027"^^<http://www.w3.org/2001/XMLSchema#gYear>
<a href="http://dbpedia.org/resource/Huantsán">http://dbpedia.org/resource/Huantsán</a>	"Huantsán"@en	6369.0		"1952"^^<http://www.w3.org/2001/XMLSchema#gYear>
<a href="http://dbpedia.org/resource/Chacraraju">http://dbpedia.org/resource/Chacraraju</a>	"Chacraraju"@en	6108.0		"1956"^^<http://www.w3.org/2001/XMLSchema#gYear>
<a href="http://dbpedia.org/resource/Fitz_Roy">http://dbpedia.org/resource/Fitz_Roy</a>	"Fitz Roy"@en	3405.0		"1952"^^<http://www.w3.org/2001/XMLSchema#gYear>

## Question 7 : solution

```
SELECT ?m ?nom ?altitude ?pnom ?année WHERE {  
  ?m rdf:type dbo:Mountain;  
    dbo:firstAscentPerson dbr:Lionel_Terray;  
    dbo:elevation ?altitude ;  
    rdfs:label ?nom.  
  FILTER langMatches( lang(?nom), "EN" )  
  OPTIONAL {  
    ?m dbo:firstAscentPerson ?p.  
    ?p rdfs:label ?pnom.  
    FILTER langMatches( lang(?pnom), "EN" )  
    FILTER (dbr:Lionel_Terray != ?p)  
  }  
  OPTIONAL {  
    ?m dbo:firstAscentYear ?année.  
  }  
} ORDER BY DESC (?altitude)
```

## Question 7 a): solution

SPARQL | HTML5 table

m	nom	altitude	pnom	année
<a href="http://dbpedia.org/resource/Makalu">http://dbpedia.org/resource/Makalu</a>	"Makalu"@en	8481.0	"Jean Couzy"@en	"1955"^^<http://www.w3.org/2001/XMLSchema#gYear>
<a href="http://dbpedia.org/resource/Chomo_Lonzo">http://dbpedia.org/resource/Chomo_Lonzo</a>	"Chomo Lonzo"@en	7804.0	"Jean Couzy"@en	"1954"^^<http://www.w3.org/2001/XMLSchema#gYear>
<a href="http://dbpedia.org/resource/Kumbhakarna_Mountain">http://dbpedia.org/resource/Kumbhakarna_Mountain</a>	"Kumbhakarna Mountain"@en	7710.0		"0027"^^<http://www.w3.org/2001/XMLSchema#gYear>
<a href="http://dbpedia.org/resource/Huantsán">http://dbpedia.org/resource/Huantsán</a>	"Huantsán"@en	6369.0		"1952"^^<http://www.w3.org/2001/XMLSchema#gYear>
<a href="http://dbpedia.org/resource/Chacaraju">http://dbpedia.org/resource/Chacaraju</a>	"Chacaraju"@en	6108.0		"1956"^^<http://www.w3.org/2001/XMLSchema#gYear>
<a href="http://dbpedia.org/resource/Fitz_Roy">http://dbpedia.org/resource/Fitz_Roy</a>	"Fitz Roy"@en	3405.0		"1952"^^<http://www.w3.org/2001/XMLSchema#gYear>

## Question 7 b):

Donnez une requête qui vous permet de trouver les sommets dont la première ascension a été effectuée par Lionnel Terray, l'année de cette première ascension et si DBpedia dispose des données le nom **et le pays de naissance** du ou des alpinistes avec qui Lionel Terray a accompli cette première.

sommet	annee	compagnonCordee	pays
"Makalu"@en	"1955"^^<http://www.w3.org/2001/XMLSchema#gYear>	"Jean Couzy"@en	"France"@en
"Chomo Lonzo"@en	"1954"^^<http://www.w3.org/2001/XMLSchema#gYear>	"Jean Couzy"@en	"France"@en
"Huantsán"@en	"1952"^^<http://www.w3.org/2001/XMLSchema#gYear>		

## Question 7

```
SELECT ?m ?nom ?altitude ?année ?pnom ?nomPays
WHERE {
  ?m rdf:type dbo:Mountain;
      dbo:firstAscentPerson dbr:Lionel_Terray;
      dbo:elevation ?altitude ;
      rdfs:label ?nom.
      FILTER langMatches( lang(?nom), "EN" )
OPTIONAL {
  ?m dbo:firstAscentPerson ?p.
  ?p rdfs:label ?pnom;
      dbo:birthPlace ?lieuNaissance.
  ?lieuNaissance dbo:country ?pays.
  ?pays rdfs:label ?nomPays.
  FILTER (dbr:Lionel_Terray != ?p)
  FILTER langMatches( lang(?pnom), "EN" )
  FILTER langMatches( lang(?nomPays), "EN" )
}
OPTIONAL {
  ?m dbo:firstAscentYear ?année.
}
} ORDER BY DESC(?altitude )
```